Crystallographic

```
> restart:
> with(PIR): with(mylinalg):  with(homalg):
> RPP:=`PIR/homalg`;
```

$$RPP := PIR/homalg$$

```
> `homalg/default`:=`PIR/homalg`;
```

$$homalg/default := PIR/homalg$$

```
> var:=[];
```

$$var := []$$

```
> Pvar(var);
```

$$[\text{“Z”}]$$

A non-faithful representation of the group $G = \langle a, b, c \mid a^2, b^2, c^2, (ab)^4, (ac)^4, (bc)^2\rangle$. This representation turns $L := \mathbb{Z}^{2\times 1}$ into a $\mathbb{Z}G$-module. We want to prove that G is infinite:

```
> Delta:=[matrix([[-1,0],[0,1]]),matrix([[0,1],[1,0]]),matrix([[0,-1],[
-1,0]])];
```

$$\Delta := [\begin{bmatrix} -1 & 0 \\ 0 & 1 \end{bmatrix}, \begin{bmatrix} 0 & 1 \\ 1 & 0 \end{bmatrix}, \begin{bmatrix} 0 & -1 \\ -1 & 0 \end{bmatrix}]$$

```
> Orbits(Delta,[Delta[1]],var);
```

$$[[\begin{bmatrix} -1 & 0 \\ 0 & 1 \end{bmatrix}, \begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}, \begin{bmatrix} 0 & 1 \\ -1 & 0 \end{bmatrix}, \begin{bmatrix} 0 & -1 \\ 1 & 0 \end{bmatrix}, \begin{bmatrix} 0 & 1 \\ 1 & 0 \end{bmatrix}, \begin{bmatrix} 0 & -1 \\ -1 & 0 \end{bmatrix}, \begin{bmatrix} 1 & 0 \\ 0 & -1 \end{bmatrix}, \begin{bmatrix} -1 & 0 \\ 0 & -1 \end{bmatrix}]]$$

```
> Delta1:=map(a->DiagMat(a,[[1]]),Delta);
```

$$\Delta 1 := \left[\begin{bmatrix} -1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}, \begin{bmatrix} 0 & 1 & 0 \\ 1 & 0 & 0 \\ 0 & 0 & 1 \end{bmatrix}, \begin{bmatrix} 0 & -1 & 0 \\ -1 & 0 & 0 \\ 0 & 0 & 1 \end{bmatrix}\right]$$

The degree of the representation Δ an the number of generators of G:

```
> n:=RPP[NumberOfGenerators](Delta[1]); m:=nops(Delta);
```

$$n := 2$$
$$m := 3$$

```
> x:=matrix(n,m);
```

$$x := \text{array}(1..2,\ 1..3,\ [])$$

```
> v:=map(op,NormalizeInput(Involution(x)));
```

$$v := [x_{1,1},\ x_{2,1},\ x_{1,2},\ x_{2,2},\ x_{1,3},\ x_{2,3}]$$

```
> X:=map(a->linalg[blockmatrix](2,2,[IdentityMap(n,var),RPP[CertainColu
mns](x,[a..a]),[RPP[Zero](n)],[RPP[One]]]),[$1..m]);
```

$$X := \left[\begin{bmatrix} 1 & 0 & x_{1,1} \\ 0 & 1 & x_{2,1} \\ 0 & 0 & 1 \end{bmatrix}, \begin{bmatrix} 1 & 0 & x_{1,2} \\ 0 & 1 & x_{2,2} \\ 0 & 0 & 1 \end{bmatrix}, \begin{bmatrix} 1 & 0 & x_{1,3} \\ 0 & 1 & x_{2,3} \\ 0 & 0 & 1 \end{bmatrix}\right]$$

```
> GA:=map(a->Compose(X[a],Delta1[a],var),[$1..m]);
```

$$GA := \left[\begin{bmatrix} -1 & 0 & x_{1,1} \\ 0 & 1 & x_{2,1} \\ 0 & 0 & 1 \end{bmatrix}, \begin{bmatrix} 0 & 1 & x_{1,2} \\ 1 & 0 & x_{2,2} \\ 0 & 0 & 1 \end{bmatrix}, \begin{bmatrix} 0 & -1 & x_{1,3} \\ -1 & 0 & x_{2,3} \\ 0 & 0 & 1 \end{bmatrix}\right]$$

Compute the derivations (satisfying the relations of the group G):

```
> rel:=[evalm(GA[1]^2), evalm(GA[2]^2), evalm(GA[3]^2),
evalm((GA[1]&*GA[2])^4), evalm((GA[1]&*GA[3])^4),
evalm((GA[2]&*GA[3])^2)];
```

$$rel := \left[\begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 2\,x_{2,1} \\ 0 & 0 & 1 \end{bmatrix}, \begin{bmatrix} 1 & 0 & x_{2,2}+x_{1,2} \\ 0 & 1 & x_{2,2}+x_{1,2} \\ 0 & 0 & 1 \end{bmatrix}, \begin{bmatrix} 1 & 0 & -x_{2,3}+x_{1,3} \\ 0 & 1 & -x_{1,3}+x_{2,3} \\ 0 & 0 & 1 \end{bmatrix}, \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}, \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}, \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}\right]$$

```
> der:=map(a->op(map(op,NormalizeInput(RPP[CertainColumns](RPP[CertainR
ows](a,[1..n]),[n+1..n+1])))),rel);
```

$$der := [0,\ 2\,x_{2,1},\ x_{2,2}+x_{1,2},\ x_{2,2}+x_{1,2},\ -x_{2,3}+x_{1,3},\ -x_{1,3}+x_{2,3},\ 0,\ 0,\ 0,\ 0,\ 0,\ 0]$$

The kernel of ψ is the group of derivations:

```
> psi:=Involution(linalg[jacobian](der,v));
```

$$\psi := \begin{bmatrix} 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 2 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 1 & 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 1 & 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 1 & -1 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & -1 & 1 & 0 & 0 & 0 & 0 & 0 & 0 \end{bmatrix}$$

Now compute the inner derivations:

```
> trv:=map(a->Compose(evalm(X[1]^(-1)),Compose(a,X[1],var),var),Delta1)
;
```

$$trv := \left[\begin{bmatrix} -1 & 0 & -2\,x_{1,1} \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}, \begin{bmatrix} 0 & 1 & x_{2,1}-x_{1,1} \\ 1 & 0 & x_{1,1}-x_{2,1} \\ 0 & 0 & 1 \end{bmatrix}, \begin{bmatrix} 0 & -1 & -x_{2,1}-x_{1,1} \\ -1 & 0 & -x_{2,1}-x_{1,1} \\ 0 & 0 & 1 \end{bmatrix}\right]$$

```
> inn:=map(b->op(op(NormalizeInput(Involution(RPP[CertainColumns](RPP[C
ertainRows](b,[1..n]),[n+1..n+1]),var)))),trv);
```

$$inn := [-2\,x_{1,1},\ 0,\ x_{2,1}-x_{1,1},\ x_{1,1}-x_{2,1},\ -x_{2,1}-x_{1,1},\ -x_{2,1}-x_{1,1}]$$

The image of ϕ is the group of inner derivations:

```
> phi:=Involution(linalg[jacobian](inn,v),var);
```

$$\phi := \begin{bmatrix} -2 & 0 & -1 & 1 & -1 & -1 \\ 0 & 0 & 1 & -1 & -1 & -1 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \end{bmatrix}$$

The first cohomology group $H^1(G,L) \cong \mathbb{Z}/2\mathbb{Z} \oplus \mathbb{Z}$. The appearance of a non-trivial free part proves that G is infinite:

```
> DefectOfHoms(phi,psi,var);
```

[[[1, 0] = [−1, 0, 0, 0, −1, −1], [0, 1] = [0, 0, 0, 0, 1, 1]], [[2, 0]], "Presentation", [2, 0], 1]

Author: Mohamed Barakat

Date:	2007-06-28
Last modified:	2007-07-22 14:35

References

[Bar07a] Mohamed Barakat, PIR: *A tiny* homalg *ring package for* Maple*-built-in principal ideal rings*, 2004-2007, (http://wwwb.math.rwth-aachen.de:8040).

[Bar07b] ______, `mylinalg`: *A linear algebra supplement package for* Maple, 2005-2007, (http://wwwb.math.rwth-aachen.de:8040).

[BR] Mohamed Barakat and Daniel Robertz, `homalg` – *A meta-package for homological algebra*, submitted. `arXiv:math.AC/0701146` and (http://wwwb.math.rwth-aachen.de/homalg).

[BR07] ______, `homalg` *project*, 2004-2007, (http://wwwb.math.rwth-aachen.de/homalg).

Lehrstuhl B für Mathematik, RWTH-Aachen University, 52062 Germany
E-mail address: mohamed.barakat@rwth-aachen.de